
I - Introduction

Le FAAC est un système industriel moderne, architecturé  autour d’un micro-contrôleur qui gère tout le processus de commande et de contrôle. Un micro-contrôleur est un composant programmable qui se présente sous forme d’une « boite noire » quelque peu austère mais d’apparence seulement : en effet, sa facilité d’emploi (si si !), sa grande polyvalence, et son prix aujourd’hui très bas fait de lui un élément incontournable des systèmes modernes en cette fin de 20ème siècle. Le micro-contrôleur remplace avantageusement  les systèmes analogiques ou en logique câblée : on n’y gagne en terme de performance, de fiabilité (moins de composants = moins de pannes), d’espace (peu volumineux), de coût au niveau de la mise en œuvre, de la maintenance ou de l’adaptation aux spécifications nouvelles (il est beaucoup plus facile et rentable de modifier un  programme que de refaire une carte électronique !)

Ne disposant pas de système de développement pour le micro-contrôleur du FAAC, nous avons utilisé les micro-contrôleurs 68HC11 et 8051 pour commander le portail.  (HOAREAU Bertrand pour le 68HC11 de Motorola et VIMBOULY Joël pour l’Intel 8051 ).
II - Description du fonctionnement

GRAFCET DE FONCTIONNEMENT

[image: image1.png]IPLLSION COMMANDE

OLVRR
PORTAL

PORTAIL OUVERT
(idouy = 1)





La demande d’ouverture sera faite soit par la télécommande (signal codé), soit par l’entrée au clavier du code secret sur 4 digits. 

L’ouverture et la fermeture du portail sont temporisées : il n’y pas de capteur de fin de course qui détecte si le portail est ouvert ou fermé. L’ordre d’ouverture est donné pendant un temps TCmax suffisant pour que le portail soit complètement ouvert. Idem pour la fermeture. 

Remarque 1 :  Lorsqu’on entre le code secret (1122), le portail s’ouvre puis se referme automatiquement au bout d’un certain temps T1. Avec le mode télécommande, il faut une impulsion pour ouvrir et une impulsion pour fermer. 

Remarque 2 : TC (tempo_courante) va s’incrémenter (TC++) lors de l’ouverture et se décrémenter (TC--) lors de la fermeture du portail. A tout moment, le système connaît la position courante du portail qui est proportionnelle à la valeur de TC ce qui permet d’optimiser la commande du moteur. (on commande le moteur juste le temps nécessaire pour le fermer ou l’ouvrir, et ce quelle que soit sa position courante)

PRINCIPE DE LA COMMANDE MOTEUR DE LA CARTE FAAC

* Le moteur 


Le moteur est du type asynchrone et fonctionne en biphasé : L'alimentation secteur étant monophasée, la deuxième phase est fabriquée par l'introduction d'un condensateur de 25uF 250V qui déphase l'alim secteur de 90°. Ce déphasage est nécessaire pour obtenir un couple moteur conséquent.

* L'inversion du sens de rotation


Pour inverser le sens de rotation du moteur il suffit de permuter l'alimentation des 2 phases. 

* Séquencement des relais


Afin d'éviter la déterioration prématurée des relais, il est nécessaire de couper l'alimentation (Relais RE3) avant de changer le sens de rotation (Relais RE1) . Il faut donc en tenir compte lors de la gestion du séquencement des relais lors des phases d'ouverture et de fermeture du portail :

Ouverture : 

ACTIVER RE3  (ALIM ON)

Tempo Ouverture  (10 secondes)

DESACTIVER RE3

Fermeture :

ACTIVER RE1  (SENS FERMETURE)

Tempo DT (quelques dixièmes de secondes)

ACTIVER RE3 (ALIM  ON)

Tempo fermeture (12 secondes)

DESACTIVER RE3 (ALIM Off)

Tempo DT (quelques dixièmes de secondes)

DESACTIVER RE1 (sens ouverture par défaut) 

 Lors d'une fermeture suivie d'une demande prématurée de réouverture (présence obstacle ou ordre télécommande) , le programme devra bien-sûr toujours respecter  ce séquencement sous peine de mauvaises surprises … (il faut désactiver RE3 , désactiver RE1 puis réactiver RE3 )

III - Mise en œuvre des cartes de développement

1. Intégration dans le système

2. Schémas structurels des cartes de développement et l’interface de puissance

IV - Programmation des microcontrôleurs

1. Microcontrôleur 68HC11

a - Présentation du programme basic11

Le programme se décompose en 5 parties principales, et gère le processus d’acquisition des données et de commande, de la même façon que le gérerait un système industriel tel le TSX17 par exemple (automate à base de 8086) ou le micro-contrôleur du FAAC

PARTIE 1

Phase d’initialisation

Répéter

   ‘ PARTIE 2 

Lecture des entrées 

   ‘ PARTIE 3
   Mise à jour des réceptivités

   ‘ PARTIE 4
   Mise à jour des bits d‘étapes

   ‘ PARTIE 5
  Activation des sorties en fonction des équations de fonctionnement

Toujours

A cela, il faut ajouter des fonctions pour l’acquisition du code secret au clavier, la gestion des temporisations temps réel non bloquantes et enfin pour réaliser une petite tempo « classique » (bloquante).

'********************************************

'*             PORTAIL FAAC                                           *

'*   PILOTE PAR  MICROCONTROLEUR 68HC11   *

'*           Programme réalisé par :                                    *

'*          BERTRAND HOAREAU                                  *

'*  Professeur Certifié Génie Electrique                          *

'*            OCTOBRE 1998                                              *

'*   Contact Email: creole@guetali.fr                             *

'********************************************

' Fichier de config E/S

#include "start.bas"

int i, s, seca, seconde, t_up_down

int  tp(4), tp_ok(4), tmax(4), nb_tempos, tp_courant, tp_courant_max

byte telecommande,barriere,code_secret,X(4), last_telecommande,commande 

byte demande_ouverture, fin_ouverture, fin_fermeture, fin_tempo1, fin_tempo2

byte presence_obstacle

byte ouvrir_portail, fermer_portail, attente , nb,clignote,VI

byte flag1,flag2,flag3, flag4 

byte code, tab(5),bon_code(5),code_index,code_ok

byte tp_ok_courant

code_index = 0

bon_code(1)= `1`

bon_code(2)= `1`

bon_code(3)= `2`

bon_code(4)= `2`

 DDRD  = 0


' PD5=PD4=PD3=PD2=input

 SCONF = 0x4C


' B,C = sorties

 lcdinit()

 PORTB = 0                      ' Arret moteur et clignotant

 print "Portail Hoareau"   'Affichage LCD Controlboy 3

 tempo (500)

 PACTL.1 = 0

' selectionner la vitesse

 PACTL.0 = 0

 TMSK2.6 = 1

' declencher le timer

 cli


' autoriser les inter

'initialisations diverses

flag1 =0

flag2 =0

flag3 =0

flag4 =0

t_up_down = 0     ' incrémente(1) ou décrémente (-1) tempo courante

tp_ok_courant = 0 ' validation temporisation

tp_courant = 0    ' Tempo courante

' Initialisation grafcet et tempos

 X(0) = 1

 X(1) = 0

 X(2) = 0

 X(3) = 0

 tmax(0) = 0

 tmax(1) = 2500  

 tmax(2) = 500    ' laps de temps min entre 2 commandes

 tmax(3) = 2500   ' non utilisé

 tp_courant_max = 2500 ' valeur max tempo courante

 nb_tempos = 3    ' (+1)

 code_secret = 0

 last_telecommande = 0

do

 ' lecture des entrées

 LDAA PORTD         

 ANDA #32            ' MASQUAGE

 EORA #32            ' inversion du bit car 

 STAA commande   ' Interrupteur carte actif niveau 0

 LDAA PORTA

 ANDA #2             ' bit1 = 1 si non actif 

 EORA #2             ' Inversion bit

 STAA barriere

  ' mise à jour des réceptivités

 ' détection front montant sur télécommande

 if last_telecommande = 0 and commande > 0 then

    telecommande = 1

 else 

    telecommande = 0

 end if 

 demande_ouverture = telecommande or barriere or code_secret

 last_telecommande = commande

 if tp_courant >= tp_courant_max then 

    fin_ouverture = 1

    fin_fermeture = 0  

 else

    if tp_courant <=1 then

       fin_ouverture = 0

       fin_fermeture = 1

    else

       fin_ouverture = 0

       fin_fermeture = 0

    end if

 end if

 fin_tempo1 = testtempo(1)

 fin_tempo2 = testtempo(2)

 presence_obstacle = barriere

 ' mise à jour des étapes

if X(0)=1 AND demande_ouverture > 0 then

   X(1)=1

   X(0)=0

end if

if X(1) = 1 then

   t_up_down = 1  ' pour incrémenter compteur

end if

if X(1)=1 AND (fin_ouverture = 1 or (telecommande > 0 and fin_tempo2=1)) then

   X(2)=1

   X(1)=0

   fin_tempo2 = 0

   tp_ok(2) = 0

   tp(2) = 0

end if

if X(2)=1 AND presence_obstacle=0 AND ((telecommande > 0 and fin_tempo2=1) or (fin_tempo1=1 and code_secret=1)) then

   X(3)=1

   X(2)=0 

   fin_tempo2 = 0

   tp_ok(2) = 0

   tp(2) = 0

end if

if X(3) = 1 then

 t_up_down = -1  ' pour décrémenter compteur

end if

if X(3)=1 AND fin_fermeture = 1 then

   X(0)=1

   X(3)=0

end if

if X(3)=1 AND presence_obstacle = 1 or (telecommande > 0 and fin_tempo2=1) then

   X(1)=1

   X(3)=0

   fin_tempo2 = 0

   tp_ok(2) = 0

   tp(2) = 0

end if

' EQUATION DES SORTIES et tempo

ouvrir_portail = X(1)

fermer_portail = X(3)

attente = X(2)

clignote = X(1) or X(3)

tp_ok(1) = X(2)  ' tempo d'attente mode automatique

tp_ok(2) = X(1) or X(2) or X(3)

tp_ok_courant = X(1) or X(3)

' Note HOAREAU B : les tempos ne doivent pas être bloquantes !

' affectation des sorties

if ouvrir_portail <> 0 and flag1=0 then

  print " ouvrir portail "

  flag1 = 1

  flag3 = 0

  flag4 = 0

  PORTB   = 1 'MOTEUR SENS 1

end if

if fermer_portail <> 0 and flag2=0 then

   print " fermer portail "

   flag2 = 1

   flag1 = 0

   PORTB = 2 ' MOTEUR SENS 2

end if

if attente <> 0  and flag3=0 then

   print "    attente     "

   flag3 = 1

   flag2 = 0

   PORTB = 0   ' éteindre clignotant et stop moteur

end if

if X(0) = 1 then

  code_secret = lirecode()  'Code via clavier du FAAC

end if

if X(0) = 1 and flag4=0 then

  flag4 = 1

   PORTB = 0   ' éteindre clignotant et stop moteur

  print "  Entrer code !  "

end if

' clignotant

if clignote <> 0 then

  if seconde<> s then

   if PORTB.2 = 1 then 

      PORTB.2 = 0

   else 

      PORTB.2 = 1

   end if

  end if

  s = seconde 

end if

loop     ' REBOUCLER VERS DO 

' ****** LES FONCTIONS *********

function testtempo(n)

  if tp(n) >= tmax(n) then

     return 1

  else

     return 0

  end if

end function

' LIRE UN CODE AU CLAVIER

function lirecode()

int retour

retour = 0

   code = keyget()

   if code <> 0 then

      putchar(code)

      code_index = code_index + 1

      tab(code_index)=code

     if code_index = 4 then

        code_index = 0

        code_ok = 1        ' VRAI

        for i=1 to 4

            if tab(i) <> bon_code(i) then

               code_ok = 0  ' FAUX

            end if               

        next i

        if code_ok = 1 then

           retour = 1   ' variable retournée

        else

          print "  CODE ERRONE !  " 

          tempo(2000) 

          print "   Entrer Code   "

        end if

        code_ok=0

     end if

   end if

   return retour  

end function

' Tempo bloquante

function tempo(cnt)


int t


for cnt=cnt to 0 step -1



for t=0 to 200









next


next


end function

' Fonction d'interruption pour Tempo non bloquante

interrupt function temporiser3 at $FFF0

   if clignote > 0 then

      seca = seca +1

     if seca > 200 then  '244 pour avoir seconde 

       seconde = seconde + 1 ' donc un peu moins d'une seconde ...

       seca = 0

     end if

   else

     seca = 0

     seconde = 0

     s = 0

   end if

  if tp_ok_courant = 1 then

     tp_courant = tp_courant + t_up_down

     if tp_courant > tp_courant_max then

        tp_courant = tp_courant_max

     end if

     if tp_courant < 1 then 

        tp_courant = 1

     end if

  end if

   for nb = 0 to nb_tempos

    if tp_ok(nb) = 1 then

       tp(nb) = tp(nb) + 1

    else 

       tp(nb) = 0

    end if

  next

  TFLG2.6 = 1 '
  autoriser l'interruption à nouveau

  end function

' Fichier ressources pour afficheur et clavier

#include "lcd.bas"

b - Explication du programme

Notons tout d’abord que le BASIC11 permet l’intégration d’instructions en assembleur directement dans le code source, ce qui est très pratique. Lors de la compilation, tout est traduit en code machine. 

Au tout début,  après un RESET hard, le programme doit initialiser le TIMER et autoriser les interruptions . Une interruption est générée par le timer toutes les 4,1ms (les bits 0 et 1 du registre PACTL sont mis à 0 . Cf documentation technique du Controlboy)   

Tous les 4,1ms, le micro-contrôleur interrompt la tache en cours, sauvegarde sur la pile tous les registres systèmes (A,B, CCR …) puis va lire à l’adresse $FFF0 l’adresse de la sous-routine d’interruption, c-a-d ici la fonction  « temporiser3 » . C’est dans cette routine que nous incrémentons les différents compteurs du programme. L’autorisation d’incrémentation est donnée par une variable de type tableau tp_ok(nb)  , nb représentant le numéro de  compteur à valider.  Cette variable est  gérée par le programme principal, et permet de synchroniser un compteur avec l’étape  voulue. 

A la fin de la fonction « temporiser3 » , les registres systèmes sont restaurés de la pile et le programme reprend là où il s’était arrêté.

Il est impératif de réaliser une temporisation  non bloquante. En effet, il faut pouvoir réagir aux entrées à tout moment et agir en conséquence (par exemple lors de la fermeture du portail, si le capteur détecte la présence d’une voiture, on rouvre le portail !) Le cas échéant, vous fabriquez un broyeur automatique …

Le programme principal va lire les entrées TOR du PORTA où sont connectées le capteur, l’interface télécommande. Les réceptivités sont rafraîchies (test des compteurs,  présence_obstacle, demande_ouverture …) 

Les bits d’étape sont ensuite mis à jour en fonctions des étapes actives et des réceptivités vraies. 

Puis après lecture des équations de sorties, on affecte les sorties correspondantes (bit 0 et 1 du  PORTB pour les deux sens moteurs, bit 2 pour le voyant clignotant) 

Les « flags » (flag1, flag2 …) permettent d’exécuter les « PRINT » 1 seule fois.  En effet, le grafcet étant lu en permanence (rebouclage avec do… loop),  l’affichage se mettrait à défiler ou  à clignoter désagréablement. 

Lorsque l’étape initiale est active, la fonction «lirecode »  va scruter régulièrement le clavier et stocker les touches dans un tableau. Après la 4ème touche, la fonction teste le code ainsi entré et renvoie « 1 » si c’est ok. Dans le cas contraire, un message « code erroné » est affiché et la fonction renvoie 0.  

L’ouverture et la fermeture du portail sont gérées par une tempo spéciale « tp_courante » qui va soit s’incrémenter (ouverture), soit se décrémenter (fermeture). Cela permet de piloter le moteur juste le temps nécessaire à la manœuvre (temps d’ouverture et fermeture proportionnel à l’entrebâillement) 

L’utilisation d’une interruption pour la gestion de l’horloge permet de réaliser des tempos très précises et indépendantes de la longueur du programme. C’est la solution adoptée dans tous les systèmes professionnels.

2. Microcontrôleur  8051

a - Présentation du programme assembleur 8051

b - Explication du programme

V - Exploitation pédagogique sur programmes simples

1. Programme « ouverture portail »

· Fin d’ouverture avec capteur fdouv

a - Description (grafcet et/ou organigramme)

b - Basic11

' ***********************************************************************

'  Exemple de programme pour simuler l'ouverture du portail en BASIC11 (basic compilé)

'  démo HB du 09/12/1998 

'

'  LES ENTREES - SORTIES sur la carte controlboy : 

'  PORT A bit 0 : fdfer

'  PORT A bit 1 : fdouv

'  PORT B bit 0 : commande RE1 (sens ouverture/fermeture)

'  PORT B bit 1 : commande RE3 (alimentation )

' ************************************************************************

' Fichier de config E/S

#include "start.bas"

' Déclaration des variables 

BYTE X(2)     ' Déclaration du tableau  des étapes 0 à 1

BYTE fdouv    ' Déclaration variable associée au capteur fin d'ouverture

BYTE commande ' Déclaration variable associée à la télécommande (simulé  par bouton poussoir)

BYTE arreter_moteur, ouvrir_portail ' variables associées aux sorties (actions)

' Initialisation

X(0) = 1

X(1) = 0

PORTB = 0 ' Arrêt moteur

 DDRD  = 0


' PD5=PD4=PD3=PD2=input

 SCONF = 0x4C


' B,C = sorties

 lcdinit()

print " GRAFCET OUVRIR "

' PROGRAMME PRINCIPAL

DO       ' début de la boucle

' Lecture des entrées

 LDAA PORTD      ' Lecture interrupteur sur carte Controlboy   

 ANDA #32            ' MASQUAGE

 EORA #32            ' inversion du bit car 

 STAA commande   ' Interrupteur carte actif niveau 0

 LDAA PORTA     ' Lecture capteur fdouv (fin ouverture) 

 ANDA #2             ' bit1 = 1 si non actif 

 EORA #2             ' Inversion bit1 du PORT A

 STAA fdouv       

' Mise à jour des étapes

IF commande > 0  AND X(0) = 1 then 

     X(1) = 1

     X(0) = 0

END IF

IF fdouv > 0 AND X(1) = 1 THEN

    X(0) = 1

    X(1) = 0

END IF

 ' Equation des sorties

ouvrir_portail = X(1)

arreter_moteur = X(0)

' Affectation des sorties 

IF ouvrir_portail = 1 THEN

    PORTB = 2 ' Activation du relais RE3 de l'alimentation moteur . 

              ' Par défaut RE1 est ouvert (sens ouverture)

END IF

IF arreter_moteur = 1 THEN

    PORTB = 0 ' Arrêter le moteur (RE3 désactivé)

END IF

LOOP   ' Reboucler vers DO

#include "lcd.bas"

' FIN DU PROGRAMME BASIC11

c - Assembleur 8051

2. Programme « fermeture portail »

· Fin de fermeture avec capteur fdfer

a - Description (grafcet et/ou organigramme)

Du fait du séquencement des relais, le grafcet de fermeture est en réalité plus complexe :

b - Basic11

' ******************************************************

'  Exemple de programme pour simuler la fermeture du portail en BASIC11 (basic compilé)

'  démo HB du 26/02/1999 

'

'  LES ENTREES - SORTIES sur la carte controlboy : 

'  PORT A bit 0 : fdfer

'  PORT A bit 1 : fdouv

'  PORT B bit 0 : commande RE1 (sens ouverture/fermeture)

'  PORT B bit 1 : commande RE3 (alimentation )

' *******************************************************

' Fichier de config E/S

#include "start.bas"

' Déclaration des variables 

int  tp(4), tp_ok(4), tmax(4), nb_tempos, nb

BYTE X(5)     ' Déclaration du tableau  des étapes 0 à 4

BYTE fdfer    ' Déclaration variable associée au capteur fin de fermeture

BYTE commande ' Déclaration variable associée à la télécommande (simulé  par bouton poussoir)

BYTE arreter_moteur, fermer_portail ' variables associées aux sorties (actions)

BYTE v, s1,s2,s3,s4,s5,bs1       

BYTE T0_D, T1_D, bebe

' Initialisation

 PACTL.1 = 0

' selectionner la vitesse

 PACTL.0 = 0

 TMSK2.6 = 1

' declencher le timer

 cli


' autoriser les interruptions

 tmax(0) = 500  ' valeur tempo 0 

 nb_tempos = 1    ' (+1)

v = 1      

X(0) = 1

X(1) = 0

X(2) = 0

X(3) = 0

X(4) = 0

PORTB = 0 ' Arrêt moteur

 DDRD  = 0


' PD5=PD4=PD3=PD2=input

 SCONF = 0x4C


' B,C = sorties

 lcdinit()

print " GRAFCET FERMER "

' PROGRAMME PRINCIPAL

DO       ' début de la boucle

' LECTURE DES ENTREES

 LDAA PORTD      ' Lecture interrupteur sur carte Controlboy   

 ANDA #32            ' MASQUAGE

 EORA #32            ' inversion du bit car 

 STAA commande   ' Interrupteur carte actif niveau 0

 LDAA PORTA     ' Lecture capteur fdfer (fin fermeture)

 ANDA #1             ' bit0 = 1 si non actif 

 EORA #1             ' Inversion bit

 STAA fdfer       

' MISE A JOUR DES RECEPTIVITES

T0_D = testtempo(0)

'  *** MISE A JOUR DES ETAPES ***

IF commande > 0  AND X(0) = 1 then 

     s1 = 1

 ELSE   

     s1 = 0

END IF

IF T0_D = 1 AND X(1) = 1 THEN

    s2 = 1

 ELSE

    s2 = 0 

END IF

IF fdfer > 0 AND X(2) = 1 THEN

   s3 = 1

 ELSE

   s3 = 0

END IF

IF T0_D = 1 AND X(3) = 1 THEN

   s4 = 1

 ELSE

   s4 = 0    

END IF

IF v = 1 and X(4) = 1  THEN

   s5 = 1

 ELSE

   s5 = 0

END IF 

' DESACTIVATION DES ETAPES

IF s1 = 1 THEN

   X(0) = 0

END IF

IF s2 = 1 THEN

   X(1) = 0

END IF

IF s3 = 1 THEN

   X(2) = 0

END IF

IF s4 = 1 THEN

   X(3) = 0

END IF

IF s5 = 1 THEN

   X(4) = 0

END IF

' ACTIVATION DES ETAPES

X(0) = s5 OR X(0)

X(1) = s1 OR X(1)

X(2) = s2 OR X(2)

X(3) = s3 OR X(3)

X(4) = s4 OR X(4)

 ' EQUATION  DES  SORTIES

tp_ok(0) = X(1) OR X(3)  ' Validation tempo T0

' AFFECTATION DES SORTIES 

IF X(0) = 1  THEN

   PORTB = 0

END IF

IF X(1) = 1  THEN

    PORTB = 1 ' Activation RE1 : SENS fermeture 

END IF

IF X(2) = 1 THEN 

    PORTB = 3 ' RE3 = 1 (ALIM ) + RE1 =1 (sens fermeture)

END IF 

IF X(3) = 1 THEN

    PORTB = 1 ' RE3 désactivé, RE1 reste actif

END IF

IF X(4) = 1 THEN

    PORTB = 0

END IF

LOOP   ' Reboucler vers DO

function tempo(cnt)


int t


for cnt=cnt to 0 step -1



for t=0 to 200









next


next

end function

' Fonction test d'une tempo n

function testtempo(n)

  if tp(n) >= tmax(n) then

     return 1

  else

     return 0

  end if

end function

'  FONCTION D'INTERRUPTION POUR TEMPO NON BLOCANTE

interrupt function temporiser3 at $FFF0

    for nb = 0 to nb_tempos

    if tp_ok(nb) = 1 then

       if tp(nb)>20000 then  ' Limit max si erreur programme 

          tp(nb) = 20000 

        end if

       tp(nb) = tp(nb) + 1

    else 

       tp(nb) = 0

    end if

  next

  TFLG2.6 = 1 '
  autoriser l'interruption à nouveau

  end function

' Fichier routines pour afficheur LCD

#include "lcd.bas"

' FIN DU PROGRAMME BASIC11

c - Assembleur 8051

3. Programme « ouverture et fermeture »

· Fin d‘ouverture et de fermeture avec capteurs fdouv et fdfer

a - Description (grafcet et/ou organigramme)

Le grafcet détaillé correspondant au programme du 68HC11: 

b - Basic11

'

'  Exemple de programme pour piloter l'ouverture et la fermeture du portail 

'  en BASIC11 (basic compilé)

'  démo HB du 26/02/1999 

' LES ENTREES SORTIES SUR LA CARTE CONTROLBOY

' Les entrées :

' PORT A bit 0 : fdfer

' PORT A bit 1 : fdouv

'

' Les sorties :

' PORT B bit 0 : commande RE1 (sens Ouverture = 0 / fermeture =1)

' PORT B bit 1 : commande RE3 (alimentation moteur)

' Fichier de config E/S

#include "start.bas"

' Déclaration des variables 

int  tp(4), tp_ok(4), tmax(4), nb_tempos, nb

BYTE X(7)     ' Déclaration du tableau  des étapes 0 à 6

BYTE fdfer, fdouv    ' Déclaration variables associées aux capteurs 

BYTE commande ' Déclaration variable associée à la télécommande (simulé  par bouton poussoir)

BYTE arreter_moteur, fermer_portail ' variables associées aux sorties (actions)

BYTE v, s1, s2, s3, s4, s5, s6, s7        

BYTE T0_D, T1_D

' Initialisation

 PACTL.1 = 0

' selectionner la vitesse

 PACTL.0 = 0

 TMSK2.6 = 1

' declencher le timer

 cli


' autoriser les interruptions

 tmax(0) = 500  ' valeur tempo 0 

 nb_tempos = 1    ' (+1)

v = 1      

X(0) = 1

X(1) = 0

X(2) = 0

X(3) = 0

X(4) = 0

X(5) = 0

X(6) = 0

PORTB = 0 ' Arrêt moteur

 DDRD  = 0


' PD5=PD4=PD3=PD2=input

 SCONF = 0x4C


' B,C = sorties

 lcdinit()

print " G. OUVRE-FERME "

' PROGRAMME PRINCIPAL

DO       ' début de la boucle

' Lecture des entrées

 LDAA PORTD      ' Lecture interrupteur sur carte Controlboy   

 ANDA #32            ' MASQUAGE

 EORA #32            ' inversion du bit car 

 STAA commande   ' Interrupteur carte actif niveau 0

 LDAA PORTA     ' Lecture capteur fdfer (fin fermeture)

 ANDA #1             ' bit0 = 1 si non actif 

 EORA #1             ' Inversion bit

 STAA fdfer       

 LDAA PORTA     ' Lecture capteur fdouv (fin ouverture)

 ANDA #2             ' bit1 = 1 si non actif 

 EORA #2             ' Inversion bit

 STAA fdouv    

' MISE A JOUR DES RECEPTIVITES

T0_D = testtempo(0)

'**** Mise à jour des étapes ****

IF commande > 0  AND X(0) = 1 THEN 

     s1 = 1

 ELSE   

     s1 = 0

END IF

IF fdouv > 0 AND X(1) = 1 THEN

    s2 = 1

 ELSE

    s2 = 0

END IF

IF commande > 0  AND X(2) = 1 THEN 

     s3 = 1        

 ELSE   

     s3 = 0

END IF

IF T0_D = 1 AND X(3) = 1 THEN

    s4 = 1

 ELSE

    s4 = 0 

END IF

IF fdfer > 0 AND X(4) = 1 THEN

   s5 = 1

 ELSE

   s5 = 0

END IF

IF T0_D = 1 AND X(5) = 1 THEN

   s6 = 1

 ELSE

   s6 = 0    

END IF

IF v = 1 and X(6) = 1  THEN

   s7 = 1

 ELSE

   s7 = 0

END IF 

' DESACTIVATION DES ETAPES

IF s1 = 1 THEN

   X(0) = 0

END IF

IF s2 = 1 THEN

   X(1) = 0

END IF

IF s3 = 1 THEN

   X(2) = 0

END IF

IF s4 = 1 THEN

   X(3) = 0

END IF

IF s5 = 1 THEN

   X(4) = 0

END IF

IF s6 = 1 THEN

   X(5) = 0

END IF

IF s7 = 1 THEN

   X(6) = 0

END IF

' ACTIVATION DES ETAPES

X(0) = s7 OR X(0)

X(1) = s1 OR X(1)

X(2) = s2 OR X(2)

X(3) = s3 OR X(3)

X(4) = s4 OR X(4)

X(5) = s5 OR X(5)

X(6) = s6 OR X(6)

 ' Equation des sorties

tp_ok(0) = X(3) OR X(5)  ' Validation tempo T0

' Affectation des sorties 

IF X(0) = 1 THEN

   PORTB = 0

END IF

IF X(1) = 1 THEN

   PORTB = 2    ' Activation relais RE3 (RE1 inactivé = ouverture)

END IF

IF X(2) = 1  THEN

   PORTB = 0    ' Désactivation RE3 

END IF

IF X(3) = 1  THEN

    PORTB = 1 ' Activation RE1 : SENS fermeture 

END IF

IF X(4) = 1 THEN 

    PORTB = 3 ' RE3 = 1 (ALIM ) + RE1 =1 (sens fermeture)

END IF 

IF X(5) = 1 THEN

    PORTB = 1 ' RE3 désactivé, RE1 reste actif

END IF

IF X(6) = 1 THEN

    PORTB = 0  ' Désactivation RE3

END IF

LOOP   ' Reboucler vers DO

function tempo(cnt)


int t


for cnt=cnt to 0 step -1



for t=0 to 200









next


next

end function

' Fonction test d'une tempo n

function testtempo(n)

  if tp(n) >= tmax(n) then

     return 1

  else

     return 0

  end if

end function

' Fonction d'interruption pour Tempo non bloquante

interrupt function temporiser3 at $FFF0

    for nb = 0 to nb_tempos

    if tp_ok(nb) = 1 then

       if tp(nb)>20000 then  ' Limit max si erreur programme 

          tp(nb) = 20000 

        end if

       tp(nb) = tp(nb) + 1

    else 

       tp(nb) = 0

    end if

  next

  TFLG2.6 = 1 '
  autoriser l'interruption à nouveau

  end function

' Fichier routines pour afficheur LCD

#include "lcd.bas"

' FIN DU PROGRAMME BASIC11

c - Assembleur 8051

4. Programme « ouverture et fermeture » temporisée

Mise en œuvre d’une temporisation avant fermeture (capteurs aussi utilisés)

a - Description (grafcet et/ou organigramme)

Et le grafcet détaillé :

b - Basic11

'

'  Exemple de programme pour piloter l'ouverture et la fermeture du portail 

'  Avec temporisation  (fermeture automatique au bout de X secondes)

'  

'  démo HB du 26/02/1999 

'

' LES ENTREES SORTIES SUR LA CARTE CONTROLBOY

' Les entrées :

' PORT A bit 0 : fdfer

' PORT A bit 1 : fdouv

'

' Les sorties :

' PORT B bit 0 : commande RE1 (sens Ouverture = 0 / fermeture =1)

' PORT B bit 1 : commande RE3 (alimentation moteur)

' Fichier de config E/S

#include "start.bas"

' Déclaration des variables 

int  tp(4), tp_ok(4), tmax(4), nb_tempos, nb

BYTE X(7)     ' Déclaration du tableau  des étapes 0 à 6

BYTE fdfer, fdouv    ' Déclaration variables associées aux capteurs 

BYTE commande ' Déclaration variable associée à la télécommande (simulé  par bouton poussoir)

BYTE arreter_moteur, fermer_portail ' variables associées aux sorties (actions)

BYTE v, s1, s2, s3, s4, s5, s6, s7        

BYTE T0_D, T1_D  ' variables de fin de temporisation 

' Initialisation

 PACTL.1 = 0

' selectionner la vitesse

 PACTL.0 = 0

 TMSK2.6 = 1

' declencher le timer

 cli


' autoriser les interruptions

 tmax(0) = 500  ' valeur tempo T0 

 tmax(1) = 2000 ' valeur tempo T1

 nb_tempos = 2    ' (+1)

v = 1      

X(0) = 1

X(1) = 0

X(2) = 0

X(3) = 0

X(4) = 0

X(5) = 0

X(6) = 0

PORTB = 0 ' Arrêt moteur

 DDRD  = 0


' PD5=PD4=PD3=PD2=input

 SCONF = 0x4C


' B,C = sorties

 lcdinit()

print " G. O-F T "

' PROGRAMME PRINCIPAL

DO       ' début de la boucle

' Lecture des entrées

 LDAA PORTD      ' Lecture interrupteur sur carte Controlboy   

 ANDA #32            ' MASQUAGE

 EORA #32            ' inversion du bit car 

 STAA commande   ' Interrupteur carte actif niveau 0

 LDAA PORTA     ' Lecture capteur fdfer (fin fermeture)

 ANDA #1             ' bit0 = 1 si non actif 

 EORA #1             ' Inversion bit

 STAA fdfer       

 LDAA PORTA     ' Lecture capteur fdouv (fin ouverture)

 ANDA #2             ' bit1 = 1 si non actif 

 EORA #2             ' Inversion bit

 STAA fdouv    

' MISE A JOUR DES RECEPTIVITES

T0_D = testtempo(0)

T1_D = testtempo(1)

' **** Mise à jour des étapes ****

IF commande > 0  AND X(0) = 1 THEN 

     s1 = 1

 ELSE   

     s1 = 0

END IF

IF fdouv > 0 AND X(1) = 1 THEN

    s2 = 1

 ELSE

    s2 = 0

END IF

IF T1_D = 1  AND X(2) = 1 THEN 

     s3 = 1        

 ELSE   

     s3 = 0

END IF

IF T0_D = 1 AND X(3) = 1 THEN

    s4 = 1

 ELSE

    s4 = 0 

END IF

IF fdfer > 0 AND X(4) = 1 THEN

   s5 = 1

 ELSE

   s5 = 0

END IF

IF T0_D = 1 AND X(5) = 1 THEN

   s6 = 1

 ELSE

   s6 = 0    

END IF

IF v = 1 and X(6) = 1  THEN

   s7 = 1

 ELSE

   s7 = 0

END IF 

' DESACTIVATION DES ETAPES

IF s1 = 1 THEN

   X(0) = 0

END IF

IF s2 = 1 THEN

   X(1) = 0

END IF

IF s3 = 1 THEN

   X(2) = 0

END IF

IF s4 = 1 THEN

   X(3) = 0

END IF

IF s5 = 1 THEN

   X(4) = 0

END IF

IF s6 = 1 THEN

   X(5) = 0

END IF

IF s7 = 1 THEN

   X(6) = 0

END IF

' ACTIVATION DES ETAPES

X(0) = s7 OR X(0)

X(1) = s1 OR X(1)

X(2) = s2 OR X(2)

X(3) = s3 OR X(3)

X(4) = s4 OR X(4)

X(5) = s5 OR X(5)

X(6) = s6 OR X(6)

 ' Equation des sorties

tp_ok(0) = X(3) OR X(5)  ' Validation tempo T0

tp_ok(1) = X(2)

' Affectation des sorties 

IF X(0) = 1 THEN

   PORTB = 0

END IF

IF X(1) = 1 THEN

   PORTB = 2    ' Activation relais RE3 (RE1 inactivé = ouverture)

END IF

IF X(2) = 1  THEN

   PORTB = 0    ' Désactivation RE3 

END IF

IF X(3) = 1  THEN

    PORTB = 1 ' Activation RE1 : SENS fermeture 

END IF

IF X(4) = 1 THEN 

    PORTB = 3 ' RE3 = 1 (ALIM ) + RE1 =1 (sens fermeture)

END IF 

IF X(5) = 1 THEN

    PORTB = 1 ' RE3 désactivé, RE1 reste actif

END IF

IF X(6) = 1 THEN

    PORTB = 0  ' Désactivation RE3

END IF

LOOP   ' Reboucler vers DO

function tempo(cnt)


int t


for cnt=cnt to 0 step -1



for t=0 to 200









next


next

end function

' Fonction test d'une tempo n

function testtempo(n)

  if tp(n) >= tmax(n) then

     return 1

  else

     return 0

  end if

end function

' Fonction d'interruption pour Tempo non bloquante

interrupt function temporiser3 at $FFF0

    for nb = 0 to nb_tempos

    if tp_ok(nb) = 1 then

       if tp(nb)>20000 then  ' Limit max si erreur programme 

          tp(nb) = 20000 

        end if

       tp(nb) = tp(nb) + 1

    else 

       tp(nb) = 0

    end if

  next

  TFLG2.6 = 1 '
  autoriser l'interruption à nouveau

  end function

' Fichier routines pour afficheur LCD

#include "lcd.bas"

' FIN DU PROGRAMME BASIC11

c - Assembleur 8051

VI - Documents annexes pour programmation

1. Microcontrôleur 68HC11

2. Intel 8051

